



## Example

Final Status	Treatment Group			
	A		B	
	N	Percent	N	Percent
Total_N				
Study Completed				
Consent Withdrawn				
Lost to Follow Up				
Physician Removal from Study				
Other				

## Data Listing

Tx Group	Center	SubjectID	Last Phase	Age	Gender	Baseline RAPID	Last RAPID

## Reporting Data

- Plan the layout
  - Summary statistics, analysis output or data listings?
  - Content of columns/rows?
  - Order of columns/rows?
  - Font
- Use a report writing tool

## Example

ID	zSiteid
1	1
2	1
3	6
4	2
5	3
6	4
7	.....

## Proc Freq

```
proc freq;  
  tables zsiteid;  
  format zsiteid;  
run;
```

## Raw SAS Frequency Output

The FREQ Procedure

zSiteID					
zSiteID	Frequency	Cumulative Percent	Cumulative Frequency	Percent	
<i>ff</i>					
<i>fff</i>					
10	5	62.50	5	62.50	
13	2	25.00	7	87.50	
14	1	12.50	8	100.00	

## PROC REPORT

```
PROC REPORT data=xxx <option(s)>;  
COLUMN column-specification(s); </ STYLE=<style-element-  
name>  
  <[style-attribute-specification(s)]>>;  
DEFINE report-item / <usage>  
  <attribute(s)>  
  <option(s)>  
  <justification>  
  <COLOR=color>  
  <'column-header-1' <...'column-header-n'>>  
  <style>;  
RUN;
```

## Proc Report Statement

- Input data set
- Controls layout of the report
  - Box
  - Center
  - Column width
  - Missing data option

## Other Statements

### Column Statement

- Describes the arrangement of columns and column headers that span >1 column

### Define Statement

- Specifies how to use a report variable
  - location
  - appearance

## Proc Report Code (site enrollment)

```
/*ods rtf file='1:\test.rtf' style=sasweb;*/  
  
options nodate pageno=1 linesize=100 pagesize=60;  
proc report data=demos nowindows center colwidth=10 missing  
  box;  
  column zsiteid N pctn;  
  define zsiteid/'Site' group width=5;  
  define n/'Total N' width=5;  
    define pctn / 'Percent of Total' format=percent7.1;  
  title 'Total Subjects By Site';  
run;  
  
/*ods rtf close;*/
```

## Proc Report Output

Site	Total N	Percent of Total
10	5	62.5%
13	2	25.0%
14	1	12.5%

## Several Options in Proc Report

<http://www2.sas.com/proceedings/forum2008/170-2008.pdf>

<http://www2.sas.com/proceedings/sugi28/015-28.pdf>

## PROC TABULATE

## PROC TABULATE

- Display of descriptive statistics
- Tabular format of:
  - Proc means
  - Proc frequency



## Proc Tabulate Code

```
proc tabulate data=demos format=7.0 style=[font_size=2];  
  class zsiteid /missing style=[font_size=2 foreground=black  
  background=white];  
  table (zsiteid='Site' ALL="Total"),( N pctn*f=7.2)  
    /box=[label="Enrollment"  
  style=[font_size=2 foreground=black background=white]]  
  printmiss  
  misstext="0";  
  keylabel PCTN="%";  
  format zsiteid site.;  
  keyword PCTN /style=[font_size=2 foreground=black  
  background=white];  
  title "Subject Enrollment, By Site ";  
run;
```

## Output

Tabulate

Site	N	%
10	5	62.5
13	2	25.0
14	1	12.5
Total	112	100.00

Report

Site	Total N	Percent of Total
10	5	62.5%
13	2	25.0%
14	1	12.5%

## Display Ethnicity by Gender by Tx

ID	Ethnic	Gender	Tx
1	1	1	1
2	2	2	1
3	1	1	2
4	3	1	1
5	1	2	1
6	2	1	2
7	.....		

## Proc Freq

```
proc sort; by tx;  
proc freq; by tx;  
  tables ethnicity*gender;  
run;  
  
proc freq;  
  tables tx*ethnicity*gender;  
run;
```

## Proc Tabulate Code

```
ods rtf text= "Table 1.2. Number of Subjects Randomized by Ethnicity and Gender";  
proc tabulate data=demos format=7.0 style=[font_size=2] missing;  
class ethnicity gender ztreatment/missing style=[font_size=2 foreground=black  
background=white];  
table (all='Total' ethnicity=""), ztreatment=""*gender=""(N colpctn='Percent'*F=7.2)  
/box=[label="Ethnicity"  
style=[font_size=2 foreground=black background=white]] printmiss  
misstext="0";  
label gender='Gender';  
format ethnicity ethnicity.;  
format gender sex.;  
format ztreatment trt.;  
classlev ethnicity gender/style=[font_size=2 foreground=black background=white];  
keyword PCTN /style=[font_size=2 foreground=black background=white];  
run;  
ods rtf close;
```

## Tabulate Output

Ethnicity	TxA				TxB			
	Male		Female		Male		Female	
	N	Percent	N	Percent	N	Percent	N	Percent
<b>Total</b>	5	100.00	69	100.00	1	100.00	37	100.00
Hispanic/Latino	0	0.00	0	0.00	0	0.00	1	2.70
Not Hispanic/Latino	4	80.00	64	92.75	1	100.00	34	91.89
Unknown	1	20.00	5	7.25	0	0.00	2	5.41

## Class Statement

- identifies the class variables
- determines the categories that PROC TABULATE uses to calculate statistics
- Options for this statement
  - 'Missing': if not included, missing values will exclude the observation from the summary

## More Statements

### Table Statement

- Required for procedure
- describes the table to create

### Var Statement

- Identifies numeric variables to use as analysis variables.

## PROC TABULATE

```
proc tabulate data=test;  
  class xxx;  
  var xxx;  
  table xxx;  
run;
```

## PROC MEANS

```
proc sort data=test; by visit;  
proc means data=test n mean std med min max;  
  by visit;  
  var time;  
run;
```

## Proc Means Output

----- Visit =1 -----

Analysis Variable : Visit

N	Mean	Std Dev	Median	Minimum	Maximum
51	30.1372549	13.4358023	30.0000000	8.0000000	70.0000000

----- Visit =2 -----

Analysis Variable : Visit

N	Mean	Std Dev	Median	Minimum	Maximum
50	23.4200000	6.6703425	23.0000000	11.0000000	39.0000000

## PROC TABULATE Code

```
proc tabulate data=time format=7.0 style=[font_size=2];
  class visit/missing style=[font_size=2 foreground=black
    background=white];
  var time;
  table time=" *(N mean std MEDIAN MIN MAX),visit="" All="Total"
    /box=[label="Session "
    style=[font_size=2 foreground=black background=white]]
  printmiss
  misstext="0";
  classlev time/style=[font_size=2 foreground=black
    background=white];
  keyword N mean std MEDIAN MIN MAX/style=[font_size=2
    foreground=black background=white];
run;
```

## Output

Session		1	2	3	4	5	6	7	8	9	10	11	12	13	Total
Time	N	51	50	48	49	42	40	40	39	38	39	38	36	35	545
	Mean	30	23	17	17	22	22	20	19	16	21	18	17	15	20
	Std	13	7	6	7	7	11	8	5	8	6	6	7	6	9
	Median	30	23	16	16	22	18	19	17	15	20	17	15	13	18
	Min	8	11	9	2	9	8	9	11	8	10	10	6	7	2
	Max	70	39	36	36	38	53	47	35	44	34	35	42	31	70

## Multiple Variables

```
ods rtf file='I:\FIELD_Poster1acorp.rtf' style=sasweb_white;
proc tabulate data=corp format=7.0 style=[font_size=2];
class q_6 q_5 q_7 q_15 q_31/missing style=[font_size=2 foreground=black
background=white];
table (ALL="Total Respondants =4" q_6='Ownership Type')
(ALL="Total Respondants =4" q_5='#Nursing Homes' )
(ALL="Total Respondants =4" q_7='Other Clinical Trials?')
(ALL="Total Respondants =4" q_15='Any policy for Trials?')
(ALL="Total Respondants =4" q_31='Likelihood of approval?'),
(All='All Respondants')*(N colPCTN*F=7.2)
/box=[label="Demographics"
style=[font_size=2 foreground=black background=white]] printmiss
misstext=[label="0"];
keylabel colPCTN="%";
keyword colPCTN /style=[font_size=2 foreground=black background=white];
run;
ods rtf close;
```

## Output

Demographics	All Respondants	
	N	%
Total Respondants =4	4	100.00
Ownership Type		
For profit	1	25.00
For profit (Corporate)	2	50.00
Not for profit	1	25.00
Total Respondants =4	4	100.00
#Nursing Homes		
1 - 9 facilities	3	75.00
10 - 24 facilities	1	25.00
Total Respondants =4	4	100.00
Other Clinical Trials?		
Yes	2	50.00
No	2	50.00

## Continuous Variables

```
proc tabulate data=corp format=7.0 style=[font_size=2];  
  class q_31;  
  var Q_1 Q_2;  
  table Q_1  
    Q_2, (N mean std MEDIAN MIN MAX)  
    /box=[label="Questions"  
  style=[font_size=2 foreground=black background=white]]  
  printmiss  
  misstext="0";  
  keyword N mean std MEDIAN MIN MAX/style=[font_size=2  
  foreground=black background=white];  
run;
```



## Output

Questions	N	Mean	Std	Median	Min	Max
Q_1	4	28	4	26	26	33
Q_2	4	22	2	23	20	24